

NAG Library Routine Document

F08HNF (ZHBEV)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08HNF (ZHBEV) computes all the eigenvalues and, optionally, all the eigenvectors of a complex n by n Hermitian band matrix A of bandwidth $(2k_d + 1)$.

2 Specification

```

SUBROUTINE F08HNF (JOBZ, UPLO, N, KD, AB, LDAB, W, Z, LDZ, WORK, RWORK,
1                      INFO)
    INTEGER          N, KD, LDAB, LDZ, INFO
    double precision    W(N), RWORK(3*N-2)
    complex*16        AB(LDAB,*), Z(LDZ,*), WORK(N)
    CHARACTER*1      JOBZ, UPLO

```

The routine may be called by its LAPACK name ***zhbev***.

3 Description

The Hermitian band matrix A is first reduced to real tridiagonal form, using unitary similarity transformations, and then the QR algorithm is applied to the tridiagonal matrix to compute the eigenvalues and (optionally) the eigenvectors.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia <http://www.netlib.org/lapack/lug>

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

- 1: JOBZ – CHARACTER*1 *Input*
On entry: if JOBZ = 'N', compute eigenvalues only.
 If JOBZ = 'V', compute eigenvalues and eigenvectors.
Constraint: JOBZ = 'N' or 'V'.
- 2: UPLO – CHARACTER*1 *Input*
On entry: if UPLO = 'U', the upper triangular part of A is stored.
 If UPLO = 'L', the lower triangular part of A is stored.
Constraint: UPLO = 'U' or 'L'.

- 3: N – INTEGER *Input*
On entry: the dimension of the arrays W and WORK as declared in the (sub)program from which F08HNF (ZHBEV) is called. n , the order of the matrix A .
Constraint: $N \geq 0$.
- 4: KD – INTEGER *Input*
On entry: if UPLO = 'U', the number of superdiagonals, k_d , of the matrix A .
 If UPLO = 'L', the number of subdiagonals, k_d , of the matrix A .
Constraint: $KD \geq 0$.
- 5: AB(LDAB,*) – **complex*16** array *Input/Output*
Note: the second dimension of the array AB must be at least $\max(1, N)$.
On entry: the upper or lower triangle of the n by n Hermitian band matrix A .
 The matrix is stored in rows 1 to $k_d + 1$, more precisely,
 if UPLO = 'U', the elements of the upper triangle of A within the band must be stored with element A_{ij} in $AB(k_d + 1 + i - j, j)$ for $\max(1, j - k_d) \leq i \leq j$;
 if UPLO = 'L', the elements of the lower triangle of A within the band must be stored with element A_{ij} in $AB(1 + i - j, j)$ for $j \leq i \leq \min(n, j + k_d)$.
On exit: AB is overwritten by values generated during the reduction to tridiagonal form.
 The first superdiagonal and the diagonal of the tridiagonal matrix T are returned in AB using the same storage format as described above.
- 6: LDAB – INTEGER *Input*
On entry: the first dimension of the array AB as declared in the (sub)program from which F08HNF (ZHBEV) is called.
Constraint: $LDAB \geq KD + 1$.
- 7: W(N) – **double precision** array *Output*
On exit: if INFO = 0, the eigenvalues in ascending order.
- 8: Z(LDZ,*) – **complex*16** array *Output*
Note: the second dimension of the array Z must be at least $\max(1, N)$.
On exit: if JOBZ = 'V', then if INFO = 0, Z contains the orthonormal eigenvectors of the matrix A , with the i th column of Z holding the eigenvector associated with $W(i)$.
 If JOBZ = 'N', Z is not referenced.
- 9: LDZ – INTEGER *Input*
On entry: the first dimension of the array Z as declared in the (sub)program from which F08HNF (ZHBEV) is called.
Constraints:
 if JOBZ = 'V', $LDZ \geq \max(1, N)$;
 otherwise $LDZ \geq 1$.
- 10: WORK(N) – **complex*16** array *Workspace*
- 11: RWORK($3 \times N - 2$) – **double precision** array *Workspace*

12: INFO – INTEGER

Output

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO = $-i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO = i , the algorithm failed to converge; i elements of an intermediate tridiagonal form did not converge to zero.

7 Accuracy

The computed eigenvalues and eigenvectors are exact for a nearby matrix $(A + E)$, where

$$\|E\|_2 = O(\epsilon)\|A\|_2,$$

and ϵ is the *machine precision*. See Section 4.7 of Anderson *et al.* (1999) for further details.

8 Further Comments

The total number of floating-point operations is proportional to n^3 if JOBZ = 'V' and is proportional to $k_d n^2$ otherwise.

The real analogue of this routine is F08HAF (DSBEV).

9 Example

This example finds all the eigenvalues and eigenvectors of the Hermitian band matrix

$$A = \begin{pmatrix} 1 & 2-i & 3-i & 0 & 0 \\ 2+i & 2 & 3-2i & 4-2i & 0 \\ 3+i & 3+2i & 3 & 4-3i & 5-3i \\ 0 & 4+2i & 4+3i & 4 & 5-4i \\ 0 & 0 & 5+3i & 5+4i & 5 \end{pmatrix},$$

together with approximate error bounds for the computed eigenvalues and eigenvectors.

9.1 Program Text

```
*      F08HNF Example Program Text
*      Mark 21 Release.  NAG Copyright 2004.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, KDMAX
      PARAMETER        (NMAX=20,KDMAX=5)
      INTEGER          LDAB, LDZ
      PARAMETER        (LDAB=KDMAX+1,LDZ=NMAX)
      CHARACTER        UPLO
      PARAMETER        (UPLO='U')
*      .. Local Scalars ..
      DOUBLE PRECISION EERRBD, EPS
      INTEGER          I, IFAIL, INFO, J, KD, N
*      .. Local Arrays ..
      COMPLEX *16      AB(LDAB,NMAX), WORK(NMAX), Z(LDZ,NMAX)
      DOUBLE PRECISION RCONDZ(NMAX), RWORK(3*NMAX-2), W(NMAX),
```

```

+          ZERRBD(NMAX)
*      .. External Functions ..
      DOUBLE PRECISION X02AJF
      EXTERNAL          X02AJF
*      .. External Subroutines ..
      EXTERNAL          DDISNA, X04DAF, ZHBEV
*      .. Intrinsic Functions ..
      INTRINSIC          ABS, MAX, MIN
*      .. Executable Statements ..
      WRITE (NOUT,*) 'F08HNF Example Program Results'
      WRITE (NOUT,*)
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N, KD
      IF (N.LE.NMAX .AND. KD.LE.KDMAX) THEN
*
*          Read the upper or lower triangular part of the symmetric band
*          matrix A from data file
*
          IF (UPLO.EQ.'U') THEN
              READ (NIN,*) ((AB(KD+1+I-J,J),J=I,MIN(N,I+KD)),I=1,N)
          ELSE IF (UPLO.EQ.'L') THEN
              READ (NIN,*) ((AB(1+I-J,J),J=MAX(1,I-KD),I),I=1,N)
          END IF
*
*          Solve the band Hermitian eigenvalue problem
*
          CALL ZHBEV('Vectors',UPLO,N,KD,AB,LDAB,W,Z,LDZ,WORK,RWORK,INFO)
*
          IF (INFO.EQ.0) THEN
*
*              Print solution
*
              WRITE (NOUT,*) 'Eigenvalues'
              WRITE (NOUT,99999) (W(J),J=1,N)
*
              IFAIL = 0
              CALL X04DAF('General',' ',N,N,Z,LDZ,'Eigenvectors',IFAIL)
*
*              Get the machine precision, EPS and compute the approximate
*              error bound for the computed eigenvalues. Note that for
*              the 2-norm, max( abs(W(i)) ) = norm(A), and since the
*              eigenvalues are returned in ascending order
*              max( abs(W(i)) ) = max( abs(W(1)), abs(W(n)) )
*
              EPS = X02AJF()
              EERRBD = EPS*MAX(ABS(W(1)),ABS(W(N)))
*
*              Call DDISNA (F08FLF) to estimate reciprocal condition
*              numbers for the eigenvectors
*
              CALL DDISNA('Eigenvectors',N,N,W,RCONDZ,INFO)
*
*              Compute the error estimates for the eigenvectors
*
              DO 20 I = 1, N
                  ZERRBD(I) = EERRBD/RCONDZ(I)
20          CONTINUE
*
*              Print the approximate error bounds for the eigenvalues
*              and vectors
*
              WRITE (NOUT,*)
              WRITE (NOUT,*) 'Error estimate for the eigenvalues'
              WRITE (NOUT,99998) EERRBD
              WRITE (NOUT,*)
              WRITE (NOUT,*) 'Error estimates for the eigenvectors'
              WRITE (NOUT,99998) (ZERRBD(I),I=1,N)
          ELSE
              WRITE (NOUT,99997) 'Failure in ZHBEV. INFO =', INFO
          END IF

```

```

      ELSE
        WRITE (NOUT,*) 'NMAX and/or KDMAX too small'
      END IF
*
99999 FORMAT (3X,(8F8.4))
99998 FORMAT (4X,1P,6E11.1)
99997 FORMAT (1X,A,I4)
      END

```

9.2 Program Data

F08HNF Example Program Data

```

      5          2                                :Values of N and KD

      (1.0, 0.0) (2.0,-1.0) (3.0,-1.0)
                (2.0, 0.0) (3.0,-2.0) (4.0,-2.0)
                                (3.0, 0.0) (4.0,-3.0) (5.0,-3.0)
                                        (4.0, 0.0) (5.0,-4.0)
                                                (5.0, 0.0) :End of matrix A

```

9.3 Program Results

F08HNF Example Program Results

Eigenvalues

```
-6.4185 -1.4094  1.4421  4.4856 16.9002
```

Eigenvectors

	1	2	3	4	5
1	-0.2534	0.6367	-0.2560	0.0171	0.1051
	-0.0538	0.0000	0.3721	0.5500	-0.0983
2	-0.0662	-0.2578	0.5344	-0.2608	0.2516
	0.4301	0.2413	0.0000	0.4869	-0.1789
3	0.5274	-0.3039	-0.4245	-0.0399	0.4994
	0.0000	-0.3481	0.0915	0.2142	-0.1513
4	0.1061	0.3450	0.4964	-0.0253	0.5611
	-0.4981	-0.0832	-0.1546	-0.1700	0.0000
5	-0.4519	-0.2469	-0.1979	0.5614	0.4837
	0.0424	0.2634	-0.1114	0.0000	0.2509

Error estimate for the eigenvalues

```
1.9E-15
```

Error estimates for the eigenvectors

```
3.7E-16  6.6E-16  6.6E-16  6.2E-16  1.5E-16
```
